

기술 자료

Fine Phase Search

Fine Phase Search

Phase Position Offset

September 29, 2007

FINE PHASE SEARCH

Fine Phase Search 의 목적

Hall Sensor 을 이용한 Phase Search 를 하는 모터들의 거의 대부분의 사용자는 Turbo Setup 이란 프로그램을 이용하여 Phase Position Offset 을 구할 것 입니다. 이처럼 구해진 값은 Rough Phase Search(\$: 온라인 명령을 이용한)를 위하여 사용 됩니다. 그러므로 매번 Phase Search 을 할 때마다 ± 30 도의 오차가 발생(6 가지 밖에 없으므로, 자세한 내용은 아래의 Phase Position Offset 부분에서 설명을 할 것입니다.)할 수 있습니다. 이럴 경우 장거리 구동에 영향을 미칠 수가 있기 때문에, 더욱 정확한 Phase Position 을 찾기 위해 Fine Phase Search 를 수행하는 것입니다.

Fine Phase Search 방법

지금부터는 1회전당 1000cts 의 모터를 사용한다는 가정을 전제로 설명할 것입니다.

아래와 같은 방법의 Manual Phase Search 을 이용하여 정확한 Fine Phase Search 를 할 수 있습니다.

3 Phase Motor 의 경우에 lxx72 의 값이 1365 또는 683 의 값을 갖습니다.(Default : 1365)

※ !! 중요 !!

1. Manual Phase Search 전의 I129 와 I179 를 이미 임의의 P 변수에 저장한 후 Phase Search 후 다시 돌려 놓으십시오.
2. Encoder 를 기준으로 Phasing 이 되는 값이 결정되기 때문에 방향에 따라 lxx72 값이 달라 질 수 있습니다. (lxx72 = 683 이고 I7mn0 = 7 일 경우 I179 값을 반대로 적용하셔야 합니다.)

- lxx72 (Motor xx Commutation Phase Angle) = 683, I7mn0 = 3 인 경우 (또는 lxx72=1365, I7mn0=7 인 경우)

M148 = 0 (Phase Search Error Bit = 0)

#1o0

I129 = 0

I179 = 3000

(모터 정지 시까지 기다림)

M171 = 0

(약 1 초간 기다림)

I179 = 0

(약 1 초간 기다림)

#1J/

- lxx72 = 1365, I7mn0 = 3 인 경우 (또는 lxx72=683, I7mn0=7 인 경우)

M148 = 0 (Phase Search Error Bit = 0)

#1o0

I129 = 0

I179 = -3000

(모터 정지 시까지 기다림)

M171 = 0


(약 1 초간 기다림)

I179 = 0

(약 1 초간 기다림)

#1J/

위와 같은 Manual Phase Search 를 한 후, Home 명령을 하여 Index(home)를 기준으로 한 Mxx기의 값을 구한다. (Manual Phase Search 가 제대로 되었다면 반복적인 Home Search 후 Mxx기의 값은 거의 변화가 없을 것 입니다. Ex. M171 = 334)



<Phase 의 한 주기가 1000Cts 인 모터의 UVW 의 6 가지 조건>

Home Search 후 Mxx기를 HOME PLC 에 적용하여 매번 Home Search 후 Phase Position 을 정확하게 적용 할 수 있습니다

Example PLC

```
CLSOE
END GAT
DEL GAT
```

```
#define PLCStatus          P100
#define PLCError           P101
#define HomeSearchComp     M145
#define PhaseSearchErr     M148
#define InPosition         M140
#define NegativeLimit      M132
#define DesiredVelZero     M133
#define OpenLoopMode       M138
#define PhasePosition      M171
#define AccelTime          I120
#define ScurveTime         I121
#define JogSpeed           I122
#define HomeSpeed          I123
```

```
OPEN PLC 7 CLEAR
  PLCStatus = 1          ; PLC 시작 알림
  PLCError = 0          ; PLC Error 상태 알림
  HomeSearchComp = 0    ; Phase Search Error Bit 0
  AccelTime = 200       ; Home Search 하기 위한 가감속 및 속도 설정
  ScurveTime = 50
  JogSpeed = 400
  HomeSpeed = 400       ; Home Search 시의 속도 설정
  COMMAND "#1$"        ; $명령을 이용한 Rough Phasing
  I5111=(100)*8388608/I10 ; 100 msec 간 기다림
```

```

WHILE(I5111>0)
ENDWHILE
I5111 = (3000)*8388608/I10
WHILE(PhaseSearchErr =1 OR InPosition = 0)
AND (OpenLoopMode = 0)
AND (I5111 > 0)
ENDWHILE ; Phase Search Error Bit 및 In Position Bit 확인
IF (OpenLoopMode = 1 OR I5111 !>0) ; Open Loop Mode 확인 및 3 초간 TimeOut 확인
    PLCError = 1 ; PLC Error 발생 알림
    DISABLEPLC7 ; PLC Error 발생 시 PLC 7 종료
    RETURN
ENDIF
COMMAND "#1J-" ; #1 모터 Negative 방향으로 이동
I5111=(100)*8388608/I10 ; 100 msec 간 기다림
WHILE(I5111>0)
ENDWHILE
I5111=(10000)*8388608/I10
WHILE(NegativeLimit =0 OR InPosition =0 OR DesiredVelZero =0)
AND (OpenLoopMode = 0)
AND (I5111 > 0)
ENDWHILE ; Negative End Limit Bit, In Position Bit 및 Desired Velocity Zero Bit 확인
IF (OpenLoopMode = 1 OR I5111 !>0) ; Open Loop Mode 확인 및 10 초간 TimeOut 확인
    PLCError = 2 ; PLC Error 발생 알림
    DISABLEPLC7 ; PLC Error 발생 시 PLC 7 종료
RETURN
ENDIF
COMMAND "#1hm" ; #1 모터 Home Search 시작
I5111=(100)*8388608/I10 ; 100 msec 간 기다림
WHILE(I5111>0)
ENDWHILE
I5111=(20000)*8388608/I10
WHILE(InPosition = 0 OR DesiredVelZero = 0 OR HomeSearchComp = 0)
AND (OpenLoopMode = 0)
AND (I5111>0)
ENDWHILE ; In Position Bit, Desired Velocity Zero Bit 및 Home Search Complete Bit 확인
IF (OpenLoopMode = 1 OR I5111 !>0) ; Open Loop Mode 확인 및 20 초간 TimeOut 확인
    PLCError = 3 ; PLC Error 발생 알림
    DISABLEPLC7 ; PLC Error 발생 시 PLC 7 종료
    RETURN
ENDIF
I5111=(100)*8388608/I10 ; 100 msec 간 기다림
WHILE(I5111>0)
ENDWHILE
PhasePosition =334 ; Manual Phase Search 에서 구한 Mxx기 값을 대입
PLCStatus = 1 ; PLC 종료 알림
DISABLEPLC7 ; Disable PLC 7
CLOSE



```

PMAC PLOT PRO 를 이용한 Phase Position Offset 구하기

Manual Phase Search 후 PMAC PLOT PRO 를 이용한 그래프를 분석하여 Phase Position Offset 을 구할 수 있습니다. 아래의 그림들을 보시면 어렵지 않게 따라 하실 수 있을 것입니다.

- Gathering 을 하기 위한 Detail Plot 설정



우선 PMAC PLOT PRO 프로그램을 실행 한 후 Detail Plot 탭을 선택 합니다.




※위의 그림에서 Source 3 는 Uncheck 해서 체크를 제거하십시오.

프로그램 실행 후 Item to Gather 를 클릭합니다.



Address to Gather 에서 아래와 같이 두 개의 M171 과 M124 의 변수를 지정을 합니다.



두 개의 변수를 설정 후 Scaling and Processing 을 클릭 합니다.




위의 Src1, Src2 두 개의 아이템을 아래의 그림과 같이 Edit 를 합니다.



<첫 번째 M171 관련 아이템>


<두 번째 M124 관련 아이템>

※ 단, Source Address 는 PMAC 의 종류에 따라 달라질 수 있습니다.




<Edit 를 마친 후 설정 된 화면>

위의 그림과 같이 설정이 제대로 되어 있다면 Item to Plot 을 클릭하여 아래의 그림과 같이 설정을 합니다. (여기까지가 Gathering 을 하기 위한 준비 설정입니다.)



- **Gathering**


지금부터는 모터의 Phase 의 위치에 따라 Hall Sensor 의 값들이 어떻게 변하는지 직접 Gathering 을 통해 Plot 그래프를 그려볼 것입니다.



우선 Define Gather Buffer 를 클릭하여 Gathering 을 하기 위한 준비를 합니다. Define Gather Buffer 버튼을 클릭 후 모터를 적당히 느린 속도로 구동을 시킵니다. (속도가 너무 빠르면 짧은 시간 내에 많은 Phase 의 변화가 생기므로 그래프를 알아보기 힘들 것 입니다.)

Define Gather Buffer -> Begin Gathering ->End Gathering -> Upload Data ->Plot Data 순차적으로 클릭 후 Plot Data 를 클릭하면 아래와 같은 그래프가 나타납니다.

- 그래프 분석



<Gathering 후 여러 Cycle 의 Phase 위치와 Hall Sensor UVW 의 변화 그래프>


위의 그래프를 보면 알 수 있듯이 이 모터는 Phase 의 한 주기가 1000Cts 를 갖는 모터 입니다. (여태껏 가정한 모터의 스펙입니다.)

※ PMAC 에서는 Hall Sensor 의 U 상이 0 이며 V 상이 0 에서 1 로 변하는 시점이 Phase Search Initial 이 되는 시점으로 정의되어 있습니다.

- M124 -> \$78000,20 (W Flag Input Status)
- M125 -> \$78000,21 (V Flag Input Status)
- M126 -> \$78000,22 (W Flag Input Status)

이며 위의 그림에 표시된 화살표의 십자가 중심점의 Phase Search 되는 시점입니다.

위의 그래프에 표시된 십자가의 정확한 중심점을 보기 위해 Data Editor 를 이용하여 값을 확인합니다.



Data Editor 를 실행하면 위의 그래프의 데이터를 정확한 수치로 보여지며 Hall Sensor UVW 의 값이 U 상은 0 이며 V 상이 0 에서 1로 바뀌는 시점의 (10 진수로 나타내면 1 에서 3 으로 바뀌는 시점) 정확한 값을 알 수 있습니다.

- **Phase Position Offset 계산**

위의 내용을 통하여 Phase Position 은 모터의 Phase 의 한 주기내의 505Cts 의 위치라는 것을 알았습니다.

위에서 가정한 모터의 Phase 의 한 주기는 1000Cts 이며 360° 로 나타낼 경우 2.77Cts 가 1° 를 의미 합니다. 여기서 505Cts 의 위치는 $505 / 1000 * 360 = 181.8°$ 입니다.

- **lxx91 적용**

lxx91 은 Hall Sensor Read 를 위한 파라미터이며, 값의 범위는 \$800000 ~ \$FF0000 까지 입니다.

Hall Sensor Read 시의 Bit 23 은 항상 1 이므로 Bit 16 ~ 21 까지의 범위를 나타냄. 1Bit 는 $360 / 64 = 5.625$

Bit 22 는 Hall Sensor 의 검출 조건을 나타내며 정상조건일 경우 0 , 반대 조건일 경우 Bit 1 입니다.


Phase Search Position 이 505Cts 이므로 $505 / 1000 * 64 = 32.32$ 이며, 이 값을 16 진수로 변환을 하면 \$20 이 됩니다.

- **Hall Sensor 가 검출 방향이 CW 일 경우 (Bit 22 = 0)**

lxx91 의 값의 Bit 23 은 항상 1 이며 Bit 22 는 상위 8Bit 를 더하면 $\$80(\text{Bit}23 = 1) + \$20 = \$A0$ 입니다.

그러므로 lxx91 의 값은 \$A00000 이 되는 것 입니다.

Cycle.Pos	U	V	W	10진수
0	1	0	0	4
60	1	1	0	6
120	0	1	0	2
180	0	1	1	3
240	0	0	1	1
300	1	0	1	5




< Hall Sensor UVW 의 검출 방향이 CW 인 경우 (소문자 uvw 가 hall sensor 위치 임)>

- Hall Sensor 가 검출 방향이 CCW 일 경우 (Bit 22 = 1)

Hall Sensor 가 반대 검출일 경우엔 lxx91 의 Bit22 가 1 이며 이때 상위 8Bit 를 더하면 \$80(Bit 23 = 1) + \$40(Bit 22 = 1) + \$20 = \$E0 입니다. 그러므로 lxx91 의 값은 \$E00000 이 되는 것 입니다

Cycle.Poc	U	V	W	10진수
0	1	0	0	2
60	1	0	1	6
120	0	0	1	4
180	0	1	1	5
240	0	1	0	1
300	1	1	0	3



< Hall Sensor UVW 의 검출 방향이 CCW 인 경우 (소문자 uvw 가 hall sensor 위치 임)>

이 값을 이용하여 온라인 명령어인 \$명령을 이용하여 Rough Phase Search 를 하게 되는 것 입니다.